Ecuaciones de los hilos

Jose M. Goicolea, 8/5/98

```
> restart;
```

Parábola

Planteamiento de las ecuaciones

```
> assume(T0>0,q>0);
```

Ecuación diferencial:

```
> ed_par:=T.0*diff(z(x),x$2)-q;
```

$$ed\_par := T0\sim \left(\frac{\partial^2}{\partial x^2} z(x)\right) - q\sim$$

llamamos **zp(x)** a la derivada para efectuar una primera integración

```
> ed_par1:=subs(diff(z(x),x)=zp(x),ed_par);
```

$$ed\_par1 := T0\sim \left(\frac{\partial}{\partial x} zp(x)\right) - q\sim$$

```
> par.d:=int(ed_par1,x=0..X);
```

$$pard := T0\sim zp(X) - q\sim X - T0\sim zp(0)$$

Condición de borde: origen de abscisas en el vértice

```
> par.d:=subs(zp(0)=0,par.d);
```

$$pard := T0\sim zp(X) - q\sim X$$

Cambio de notación para una segunda integración

```
> par.d:=subs(X=x,zp(x)=diff(z(x),x),par.d);
```

$$pard := T0\sim \left(\frac{\partial}{\partial x} z(x)\right) - q\sim x$$

```
> par:=int(par.d,x=0..X);
```

$$par := T0\sim z(X) - \frac{1}{2} q\sim X^2 - T0\sim z(0)$$

Condición de borde: origen de ordenadas en el vértice

```
> par:=subs(z(0)=0,par);
```

$$par := T0\sim\, z(X) - \frac{1}{2} q\sim X^2$$

Volvemos a cambiar la notación y eliminamos $z(x)$

```
> par:=subs(X=x,par);
> ec_par:=z(x)=solve(par,z(x));
```

$$par := T0\sim\, z(x) - \frac{1}{2} q\sim x^2$$

$$ec\_par := z(x) = \frac{1}{2}\frac{q\sim x^2}{T0\sim}$$

Asignamos el valor de la función parábola a $z(x)$

```
> zpar:=subs(ec_par,z(x));
```

$$zpar := \frac{1}{2}\frac{q\sim x^2}{T0\sim}$$

Ahora podemos calcular, por ejemplo, la longitud de la parábola:

```
> long_par:=int(sqrt(1+diff(zpar,x)^2),x);
```

$$long\_par := \frac{\frac{1}{2} x \sqrt{T0\sim^2 + q\sim^2 x^2} + \frac{1}{2}\frac{T0\sim^2 \operatorname{arcsinh}\left(\frac{q\sim x}{T0\sim}\right)}{q\sim}}{T0\sim}$$

que admite alguna simplificación

```
> long_par:=expand(long_par);
```

$$long\_par := \frac{1}{2}\frac{x\sqrt{T0\sim^2 + q\sim^2 x^2}}{T0\sim} + \frac{1}{2}\frac{T0\sim \operatorname{arcsinh}\left(\frac{q\sim x}{T0\sim}\right)}{q\sim}$$

```
> potencial:=q*int(1/sqrt(1+(1/diff(zpar,x))^2),x);
```

$$potencial := \frac{T0\sim^2 + q\sim^2 x^2}{\sqrt{\frac{T0\sim^2 + q\sim^2 x^2}{x^2}}\, x}$$

Simplificando esta expresión,

> **potencial:=simplify(potencial,assume=positive);**

$$potencial := \sqrt{T0\sim^2 + q\sim^2 x^2}$$

Catenaria

Catenaria: carga cte. por ud. de longitud

Ecuación diferencial:

> **ed_cat:=T.0*diff(z(x),x$2)/sqrt(1+diff(z(x),x)^2)-q;**

$$ed\_cat := \frac{T0\sim \left(\frac{\partial^2}{\partial x^2} z(x)\right)}{\sqrt{1 + \left(\frac{\partial}{\partial x} z(x)\right)^2}} - q\sim$$

llamamos **zp(x)** a la derivada para efectuar una primera integración

> **ed_cat1:=subs(diff(z(x),x)=zp(x),ed_cat);**

$$ed\_cat1 := \frac{T0\sim \left(\frac{\partial}{\partial x} \mathrm{zp}(x)\right)}{\sqrt{1 + \mathrm{zp}(x)^2}} - q\sim$$

> **cat.d:=int(ed_cat1,x);**

$$catd := T0\sim \operatorname{arcsinh}(\mathrm{zp}(x)) - q\sim x$$

Lo que sigue se emplearía para hacer la integral definida

> **#cat.d:=int(ed_cat1,x=0..X);**

Condición de borde: origen de abscisas en el vértice

> **#cat.d:=subs(zp(0)=0,cat.d);**

Cambio de notación para una segunda integración

```
> #cat.d:=subs(X=x,zp(x)=diff(z(x),x),cat.d);
```

```
> readlib(isolate):
```

```
> catd:=isolate(cat.d,zp(x));
```

$$catd := \mathrm{zp}(x) = \sinh\left(\frac{q\sim\, x}{T0\sim}\right)$$

```
> catd1:=subs(zp(x)=diff(z(x),x),catd);
```

$$catd1 := \frac{\partial}{\partial x}\,\mathrm{z}(x) = \sinh\left(\frac{q\sim\, x}{T0\sim}\right)$$

```
> ec_cat:=map(int,catd1,x);
```

$$ec\_cat := \mathrm{z}(x) = \frac{T0\sim\,\cosh\left(\frac{q\sim\, x}{T0\sim}\right)}{q\sim}$$

```
> ec_cat:=subs(T0=q*a,ec_cat);
```

$$ec\_cat := \mathrm{z}(x) = a\,\cosh\left(\frac{x}{a}\right)$$

```
> zcat:=subs(ec_cat,z(x));
```

$$zcat := a\,\cosh\left(\frac{x}{a}\right)$$

Ejemplo

Enunciado

Cable de peso por ud. de longitud q , Luz L y flecha f . Calcular la configuración de equilibrio así como las tensiones mínima y máxima (T_{min}, T_{max}). Obtener la solución general y particularizar para los valores $q = 2$, $L = 200$, $f = 40$.

```
> datos:={q=2,L=200,f=40};
```

$$datos := \{L = 200, f = 40, q\sim = 2\}$$

Aproximación mediante la parábola

Particularizando en la ecuación de la parábola,

> **parn:=subs(z(x)=f,x=L/2,ec_par);**

$$parn := f = \frac{1}{8}\frac{q\sim L^2}{T0\sim}$$

de donde la Tensión mínima es

> **T.min:=isolate(parn,T0);**

$$Tmin := T0\sim = \frac{1}{8}\frac{q\sim L^2}{f}$$

La ecuación de la parábola, en función de los datos del problema resulta

> **ec_parn:=subs(T.min,ec_par);**

$$ec\_parn := z(x) = 4\frac{x^2 f}{L^2}$$

La tensión máxima es, en un extremo,

> **T.max:=sqrt(subs(T.min,T0)^2+(q*L/2)^2);**

$$Tmax := \frac{1}{8}\sqrt{\frac{q\sim^2 L^4}{f^2} + 16\, q\sim^2 L^2}$$

> **T.max:=simplify(T.max,assume=positive);**

$$Tmax := \frac{1}{8}\frac{q\sim L\sqrt{L^2 + 16 f^2}}{f}$$

> **long_par.n:=2*subs(T.min,x=L/2,long_par);**

$$long\_parn := 4\frac{f\sqrt{\frac{1}{64}\frac{q\sim^2 L^4}{f^2} + \frac{1}{4}q\sim^2 L^2}}{q\sim L} + \frac{1}{8}\frac{L^2 \operatorname{arcsinh}\left(4\frac{f}{L}\right)}{f}$$

Resultados numéricos:

> **ec_parn:=subs(datos,ec_parn);**

$$ec\_parn := z(x) = \frac{1}{250}x^2$$

> **T.min.p:=subs(datos,T.min);**

$$Tminp := T0\sim = 250$$

> **T.max.p:=T[max]=evalf(subs(datos,T.max));**

$$Tmaxp := T_{max} = 320.1562119$$

> **long_par.n:=evalf(subs(datos,long_par.n));**

$$long\_parn := 219.6460168$$

Aproximación mediante la catenaria

Al sustituir los datos resulta una ecación trascendente:

> **catn:=subs(z(x)=a+f,x=L/2,ec_cat);**

$$catn := a + f = a \cosh\left(\frac{1}{2}\frac{L}{a}\right)$$

que debe resolverse por métodos numéricos, por lo cual debemos sustituir los valores de los datos:

> **catn:=subs(datos,catn);**

$$catn := a + 40 = a \cosh\left(100\,\frac{1}{a}\right)$$

Valor del parámetro a , para el caso de la parábola (**a.p**):

> **a.p:=subs(T.min.p,T0)/subs(datos,q);**

$$ap := 125$$

Resolvemos numéricamente apra la catenaria (**a.c**), tomando como valor inicial para la iteración numérica el de la parábola (**a.p**)

> **a.c:=fsolve(catn,a=ap);**

$$ac := 131.1725174$$

de donde la Tensión mínima es

> **T.min.c:=T0=subs(datos,q*a.c);**

$$Tminc := T0\sim = 262.3450348$$

La ecuación de la catenaria, en función de los datos del problema resulta

> **ec_catn:=subs(a=a.c,ec_cat);**

$$ec\_catn := z(x) = 131.1725174 \cosh(.007623548132\,x)$$

La tensión máxima es, en un extremo,

> **T.max.c:=T[max]=q*(a+f);**

$$Tmaxc := T_{max} = q\sim(a+f)$$

> **T.max.c:=subs(datos,a=a.c,T.max.c);**

$$Tmaxc := T_{max} = 342.3450348$$

y la longitud de la misma

> **L.cat:=2*a*sinh((L/2)/a);**

$$Lcat := 2\,a\,\sinh\left(\frac{1}{2}\frac{L}{a}\right)$$

> **L.cat.n:=evalf(subs(a=a.c,datos,L.cat));**

$$Lcatn := 219.9436417$$

Dibujo de resultados

Rango del dibujo

> **rango:=subs(datos,-L/2..L/2);**

$$rango := -100\,..\,100$$

> **plot(subs(ec_parn,z(x)),x=rango,
scaling=constrained,color=red,title='parabola',thickness=2);**



> **plot(subs(ec_catn,z(x)),x=rango,
scaling=constrained,color=green,title='Catenaria',thickness=2);**



> **plot([subs(ec_catn,z(x))-a.c,subs(ec_parn,z(x))],x=rango, scaling=constrained,color=[red,green], title='comparacion_parabola_catenaria',thickness=2);**



>